
Subject: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [mahmudornob](#) on Thu, 23 Aug 2018 02:29:22 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear Altruist,

I am trying to replicate Bangladesh DHS Final Report 2014's table 9.5 (Place of Delivery & Facility-based Delivery) & 9.8 (Birth attended by Medically Trained Provider).

I have generated a syntax (please find the attached do file), following the suggestions from old posts, which shows a slight deviation from the reported values. The reason should be, my total N is 4,895 whereas DHS table N is 4,904.

Also, I could not find any variable related birthing hut as shown in table 9.5.

Would anybody please suggest me where I am making mistake? It would be an immense help.

Thanking you,
Mahmudul

File Attachments

1) [Do-file for PoD and AD.do](#), downloaded 582 times

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [Mlue](#) on Wed, 29 Aug 2018 09:02:39 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hello Mahmudul,

I am also getting 4,895. I have no idea how to get 4,904 as reported in the BDHS2014 report.

Tsawe

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [Bridgette-DHS](#) on Wed, 29 Aug 2018 12:25:35 GMT

[View Forum Message](#) <> [Reply to Message](#)

There are 9 missing cases which brings the number to 4,904 - please see the frequency below:

File Attachments

1) [M15.PNG](#), downloaded 910 times

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [Mlue](#) on Wed, 29 Aug 2018 13:42:15 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hi Bridgette,

I still can't figure out how the "9" (Missing) is derived.

I checked the report and it states that: Quote>Note: Total include six births with missing information on number of antenatal care visits..

I used a similar code to that posted by Mahmudul.

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [Bridgette-DHS](#) on Wed, 29 Aug 2018 14:24:46 GMT

[View Forum Message](#) <> [Reply to Message](#)

Following is a response from Senior Data Processing Specialist, Fidele Mutima:

Table 9.5:

Variables used to run table 9.5 are: M15 (Place of delivery) Values 10 to 19 for home; 20 to 29 for public sector; 30 to 39 for private sector; 40 to 41 for NGO; 46 for birthing hut; and 96 to 99 (missing) for other/missing. The total N is 4904; you probably might not have considered the missing.

Table 9.8:

As stated in the footnote, Medically trained provider includes doctor, nurse, midwife, paramedic, FWV (Family welfare visitor) and CSBA (Community health care provider). Variables used to run table 9.8 are: M3A = 1 for doctor; M3B = 1 for nurse/midwife/paramedic; M3C = 1 for FWV; M3D = 1 for CSBA; M3F = 1 for CHCP; S433G = 1 or S433H = 1 for HA/FWA; S433I = 1 for NGO worker; M3G = 1 for trained TBA; M3H = 1 for untrained TBA; M3I = 1 for unqualified doctor; M3L = 1 or M3K = 1 or M3M = 1 for relatives/ friends and others; M3N = 1 for no one; and all the rest are included in missing.

Variable related to Birthing Hut:

For "Place of Delivery", "Delivery Hut" was coded 36 in variable Q434 in the raw data. In the recode data, "Delivery Hut" was included in code 46 in variable M15 ("Other NGO sector").

Thank you again for using the DHS data and I hope that these responses will clarify differences you had when running tables.

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [mahmudornob](#) on Thu, 30 Aug 2018 05:30:48 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear Mlue,

Still I am getting 4,895. After keeping the cases where births occurred in the three years preceding the survey, I checked any cases for missing but didn't get any missing case.

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [mahmudornob](#) on Thu, 30 Aug 2018 05:40:24 GMT

[View Forum Message](#) <> [Reply to Message](#)

Dear Fidele Mutima and Bridgette,

Thanks for sharing your suggestions on my query. I really do appreciate your efforts. However, I am getting 4,895 still now. After keeping the cases where births occurred in the three years preceding the survey (as per reported table) by using the following syntax, I checked any cases for missing but unfortunately didn't get any missing case in the data.

```
///REPLICATING TABLE 9.5///Place of Delivery
```

```
///VARIABLES GENERATION///
```

```
cd " C:\Users\Ornia\Downloads\BD_2014_DHS STATA\BDBR72DT"
```

```
use " BDBR72FL", clear
```

```
set more off
```

```
*****
```

```
**WEIGHT VARIABLE
```

```
gen weight = v005/1000000
```

```
*****
```

```
*****
```

```
** SURVEY SET
```

```
gen psu = v021
```

```
gen strata = v023
```

```
svyset psu [pw = weight], strata(strata) vce(linearized)
```

```
*svydes
```

```
*****
```

```
** BIRTHS IN THE THREE YEARS PRECEDING THE SURVEY
```

```
keep if v238>0 & v238!=.
```

```
*****
```

```
edit
```

sort m15

Would you mind to check the publicly downloadable data?

Your suggestion would be invaluable for my research.

Thanking you,
Mahmudul

File Attachments

1) [Capture for DHS.JPG](#), downloaded 559 times

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [Mlue](#) on Thu, 30 Aug 2018 08:02:10 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hello Mahmudul,

The issue is the Stata datafile. It does not have the category for missing "99", hence it seems impossible to get these results. The SAS datafile also does not have category 99 for variable "m15".

You may request the DHS Program to check this issue. While you wait for this to be resolved, you can try the following options:

1. Download the SPSS version (if you have access to SPSS) and save the data as Stata from this software package.
2. Download and use the ASCII file in Stata
3. Download the SPSS file, and try the code below. It will work ONLY if your PC is 32-bit
4. DO NOTHING, and wait for the DHS team to resolve the issue.

/*

DOWNLOAD "SPSS" VERSION OF THE DATASET AND USE IT
BECAUSE THE Stata VERSION DOES NOT HAVE CATEGORY "99"

*/

ssc install usespss // will work on a 32-bit system

```
clear all
set matsize 800
set mem 1g
set maxvar 9000
```

```
cd "." // PATH
usespss "BDKR72FL.SAV", clear // SPSS VERSION OF THE DATASET. usespss will work only
on 32-bit system
set more off
```

```
*****
```

```
** WEIGHT VARIABLE
gen weight = V005/1000000
```

```
*****
```

```
** SURVEY SET
gen psu = V021
gen strata = V023
svyset psu [pw = weight], strata(strata) vce(linearized)
```

```
*****
```

```
rename V013 age
rename V106 education
rename V190 wealth
rename V025 residence
rename V024 region
```

```
*****
```

```
*keep if V238 >=1 & V238 <=3 // births in the last THREE years
```

```
*****
```

```
** DELIVERY CARE = Place of delivery
cap drop delivery_place
recode M15 (20/29=1 "Public sector") (30/39=2 "Private sector") ///
(40/41=3 "NGO") (46=4 "Birthing hut") (10/19=5 "Home") ///
(96/99=6 "Other/missing"), gen(delivery_place)
*tab delivery_place [iw=weight]
*svy: tab delivery_place, count format(%4.0f)
*svy: tab delivery_place, percent format(%4.1f)
```

```
*****
```

```
*****
```

```
** FACILITY DELIVERY
cap drop facility_deliveries
recode delivery_place (1/3=1 "Facility births") (4/6=0 "Non-facility"), gen(facility_deliveries)
label var facility_deliveries "Facility-based deliveries"
label val facility_deliveries facility_deliveries
```

```
*tab facility_deliveries [iw=weight], m
*svy: tab facility_deliveries, count format(%4.0f)
*svy: tab facility_deliveries, percent format(%4.1f)
```

```
*****
```

```
**
```

```
=====
```

```
**
```

** SKILLED BIRTH ATTENDANT

```
cap drop skilled_birth
gen skilled_birth=0
replace skilled_birth = 1 if M3A==1 | M3B==1 | M3C==1 | M3D==1
replace skilled_birth = 0 if M3A==0 & M3B==0 & M3C==0 & M3D==0
label define skilled_birth 0"Unskilled" 1"Skilled"
label var skilled_birth "Birth delivered by skilled birth attendant"
label val skilled_birth skilled_birth
*tab skilled_birth [iw=weight], m
*svy: tab skilled_birth, percent format(%4.1f)
*svy: tab wealth skilled_birth, percent format(%4.1f) row
```

```
**
```

```
=====
```

```
**
```

```
**
```

```
=====
```

```
**
```

** ANTENATAL CARE

```
cap drop antental_visits
recode M14 (0=0 "None") (1/3=1 "1-3") (4/20=2 "4+") (else=9 "Other/missing"),
gen(antental_visits)
label var antental_visits "Antenatal visits"
label val antental_visits antental_visits
```

```
**
```

```
=====
```

```
**
```

```
**
```

```
=====
```

```
**
```

** C-SECTION

```
cap drop c_section
recode M17 (1=1 "C-section") (else=0 "Not C-section"), gen(c_section)
label var c_section "Births delivered by a C-section"
label val c_section c_section
```

```

**
=====
**
**
=====
**

/*
cap drop skilled_births
gen skilled_births = 0
label define skilled_births 0"Unskilled" 1"Skilled"
label var skilled_births "Birth delivered by skilled birth attendant"
label val skilled_births skilled_births

** SKILLED BIRTH ATTENDANTS RECODE

foreach xvar of varlist M3A M3B M3C M3D {
replace skilled_births=1 if `xvar'==1
}
*

tab skilled_births [iw=weight], m
svy: tab wealth skilled_births, percent format(%4.1f) row
*/

*=====
=====*

keep if delivery_place !=.

*****

** CHECK
tab delivery_place [iw=weight], m
tab facility_deliveries [iw=weight], m
tab skilled_birth [iw=weight], m
tab c_section [iw=weight], m

svy: tab delivery_place, count format(%4.0f)
svy: tab facility_deliveries, count format(%4.0f)
svy: tab skilled_birth, count format(%4.0f)
svy: tab c_section, count format(%4.0f)

svy: tab delivery_place, percent format(%4.1f)
svy: tab facility_deliveries, percent format(%4.1f)
svy: tab skilled_birth, percent format(%4.1f)

```

```
svy: tab c_section, percent format(%4.1f)
```

```
exit
```

```
svy: tab wealth delivery_place, percent format(%4.1f) row  
svy: tab wealth facility_deliveries, percent format(%4.1f) row  
svy: tab wealth skilled_birth, percent format(%4.1f) row  
svy: tab wealth c_section, percent format(%4.1f) row
```

Subject: Re: Slight Deviation from Reported Value: BDHS 2014 place of delivery and assistance during delivery

Posted by [mahmudornob](#) on Thu, 30 Aug 2018 09:32:36 GMT

[View Forum Message](#) <> [Reply to Message](#)

Hello Mlue,

Thanks for your advice. You always saved my time. Without your guidance, it would be impossible for me to learn Stata in this short period of time. Now I will try option 1. However, option 4 seems like more convenient.

Thanks once again...
